#include "atomic_op_asm.h"

	.text

ENTRY(_atomic_cas_32)
	movl	4(%sp), %a0
	movl	8(%sp), %d0
	movl	12(%sp), %d1
	casl	%d0, %d1, (%a0)
	/* %d0 now contains the old value */
	movl	%d0, %a0	/* pointers return also in %a0 */
	rts
END(_atomic_cas_32)

ATOMIC_OP_ALIAS(atomic_cas_32,_atomic_cas_32)
ATOMIC_OP_ALIAS(atomic_cas_uint,_atomic_cas_32)
STRONG_ALIAS(_atomic_cas_uint,_atomic_cas_32)
ATOMIC_OP_ALIAS(atomic_cas_ulong,_atomic_cas_32)
STRONG_ALIAS(_atomic_cas_ulong,_atomic_cas_32)
ATOMIC_OP_ALIAS(atomic_cas_ptr,_atomic_cas_32)
STRONG_ALIAS(_atomic_cas_ptr,_atomic_cas_32)
CRT_ALIAS(__sync_val_compare_and_swap_4,_atomic_cas_32)

ATOMIC_OP_ALIAS(atomic_cas_32_ni,_atomic_cas_32)
STRONG_ALIAS(_atomic_cas_32_ni,_atomic_cas_32)
ATOMIC_OP_ALIAS(atomic_cas_uint_ni,_atomic_cas_32)
STRONG_ALIAS(_atomic_cas_uint_ni,_atomic_cas_32)
ATOMIC_OP_ALIAS(atomic_cas_ulong_ni,_atomic_cas_32)
STRONG_ALIAS(_atomic_cas_ulong_ni,_atomic_cas_32)
ATOMIC_OP_ALIAS(atomic_cas_ptr_ni,_atomic_cas_32)
STRONG_ALIAS(_atomic_cas_ptr_ni,_atomic_cas_32)

ENTRY(__sync_bool_compare_and_swap_4)
	movl	4(%sp), %a0
	movl	8(%sp), %d0
	movl	12(%sp), %d1
	casl	%d0, %d1, (%a0)
	beqs	1f
	clrl	%d0	/* return false */
	rts
1:	movql	#1, %d0	/* return success */
	rts
END(__sync_bool_compare_and_swap_4)

ENTRY(_atomic_cas_16)
	movl	4(%sp), %a0
	movw	8+2(%sp), %d0		/* lower word */
	movw	12+2(%sp), %d1		/* lower word */
	casw	%d0, %d1, (%a0)
	/* %d0 now contains the old value */
	rts
END(_atomic_cas_16)

ATOMIC_OP_ALIAS(atomic_cas_16,_atomic_cas_16)
CRT_ALIAS(__sync_val_compare_and_swap_2,_atomic_cas_16)

ENTRY(__sync_bool_compare_and_swap_2)
	movl	4(%sp), %a0
	movw	8+2(%sp), %d0		/* lower word */
	movw	12+2(%sp), %d1		/* lower word */
	casw	%d0, %d1, (%a0)
	/* %d0 now contains the old value */
	beqs	1f
	clrl	%d0	/* return failure */
	rts
1:	movql	#1, %d0	/* return success */
	rts
END(__sync_bool_compare_and_swap_2)

ENTRY(_atomic_cas_8)
	movl	4(%sp), %a0
	movb	8+3(%sp), %d0		/* lower byte */
	movb	12+3(%sp), %d1		/* lower byte */
	casb	%d0, %d1, (%a0)
	/* %d0 now contains the old value */
	rts
END(_atomic_cas_8)

ATOMIC_OP_ALIAS(atomic_cas_8,_atomic_cas_8)
CRT_ALIAS(__sync_val_compare_and_swap_1,_atomic_cas_8)


ENTRY(__sync_bool_compare_and_swap_1)
	movl	4(%sp), %a0
	movb	8+3(%sp), %d0		/* lower byte */
	movb	12+3(%sp), %d1		/* lower byte */
	casb	%d0, %d1, (%a0)
	/* %d0 now contains the old value */
	beqs	1f
	clrl	%d0	/* return failure */
	rts
1:	movql	#1, %d0	/* return success */
	rts
END(__sync_bool_compare_and_swap_1)

